

Making a Custom Web Page

The new generation of dataloggers have the ability to host a web page over an IP connection. This connection can be over a cell phone modem or an ethernet connection using an NL115. By default the datalogger will host a basic web page with links to the historical data stored on the logger and the station status table. This web page can be replaced with a user defined web page in the datalogger programming.

Example Web Page Code

```
Public PTemp, batt_volt
Public Commands As String * 100
Dim i

DataTable      (Test,1,1)
    DataInterval (0,15,Sec,10)
    Minimum      (1,batt_volt,FP2,0,False)
    Sample       (1,PTemp,FP2)
EndTable

WebPageBegin ("default.html",Commands)
  HTTPOut("<img src=""cpu/csa_logo.jpg"" />")
  HTTPOut("<p>Welcome to the CSA datalogger!</p>")
  HTTPOut("<p>Battery Voltage: " + batt_volt + "</p>")
  HTTPOut("<p>Watchdogs: " + Status.WatchdogErrors(1,1) + "</p>")
  HTTPOut("Last ten records from test table:<br/>")
  For i = 1 To 10
    Dim lastrecord_string As String * 256
    GetRecord (lastrecord_string,Test,i)
    HTTPOut(lastrecord_string + "<br/>")
  Next i
WebPageEnd

BeginProg
  Scan (1,Sec,0,0)
    PanelTemp (PTemp,250)
    Battery (batt_volt)
    CallTable Test
  NextScan
EndProg
```

The example code above creates a web page called default.html (this will overwrite the CR1000 default web page) which displays the following on separate lines;

- `""` - An `` html tag has been used to insert a company logo stored on the datalogger CPU drive.
- `"<p>Battery Voltage: " + batt_volt + "</p>"` - The logger battery voltage has been added onto the end of a paragraph of text (`<p></p>` tags).
- `"<p>Watchdogs: " + Status.WatchdogErrors(1,1) + "</p>"` - The number of watchdog errors has also been added on to the end of a paragraph. This time the value has come from the status table (this syntax can be used to access any status table value, refer to `Tablename.FieldName` in the CRBasic help for more information).
- `lastrecord_string + "
"` - The last ten records in the Test table are displayed one after the other using a "for loop" to avoid writing ten lines of code. All CRBasic programming functions, such as for and while loops) are available between `WebPageBegin` and `WebPageEnd`. The `GetRecord` instruction pulls data from a specified number of records back from a datatable and creates a string from that data. For example: "2010-02-25 09:31:00",0,11.87,28.14